
C=64 Mania – POKE cose sugli sprite
Giovanni A. Cignoni, Giuseppe Lettieri – Giugno 2015

Il MOS 6567 (NTSC) o 6589 (PAL) Video Interface Chip II, per tutti VIC-II, è il chip video del C=64. In 
generale, un chip video legge una parte della memoria del calcolatore e la interpreta per generare il 
segnale video da mandare al monitor o alla TV. Il tutto 25 volte al secondo (il C=64, di più oggi). I 
programmi costruiscono nella memoria quel che si vuol far vedere e il gioco è fatto.
Il VIC-II aggiunge gli sprite, oggetti grafici che pensa il VIC a posizionare sullo schermo gestendo la 
sovrapposizione e il ripristino dello sfondo. Per la realizzazione di videogiochi è un gran vantaggio: il 
6510 deve dire al VIC-II solo dove vuole lo sprite, non deve disegnarlo ogni volta nella memoria.

Uno sprite monocromatico è definito da una griglia di 24 × 21 bit (0 trasparente, 1 pieno) quindi 63 byte i 
cui valori sono determinati prendendo, per righe, otto a otto, i bit che definiscono lo sprite. Il VIC-II 
gestisce 8 sprite, numerati da 0 a7. Giusto per informazione, ci sono anche gli sprite a 4 colori, codificati 
sulla stessa griglia, prendendo i bit 2 a 2 (in pratica dimezzando la risoluzione orizzontale).

Gestire gli sprite da programma significa inserire valori nei registri del VIC-II, mappati su locazioni note 
della memoria del C=64.

Locazioni importanti

53248, base V delle locazioni del VIC
V e V+1, per impostare il colore del bordo e il colore di sfondo dello schermo
V+21, per abilitare gli sprite, un bit per ognuno, lo sprite 0 è il bit meno significativo
V+39+S, per impostare il colore dello sprite S
02040+S, per impostare il numero di blocco della bitmap dello sprite S
V+2S, per impostare gli 8 bit meno significativi della coordinata x dello sprite S 
V+2S+1, per impostare la coordinata y dello sprite S
V+16, per impostare il 9° bit della coordinata x, un bit per ognuno, lo sprite 0 è il bit meno significativo

Esempio minimo in BASIC 64
10 rem sfondo nero, bordo rosso 
15 poke 53281,0: poke 53280,2 
20 v=53248: rem base dei registri 
30 poke v+21,1: rem abilita sprite 0 
40 poke 2040,13:rem usa blocco n.13 
45 poke v+39,5: rem colore verde 
48 rem carica lo sprite nel blocco n.13 
49 rem (13*64 = 832) 
50 for n=0to63:read q:poke 832+n,q:next 
60 x = 168: y = 140:rem coordinate 
70 dx = 2:rem incremento x 
72 rem inserisce x in v e y in v+1 
74 rem se x non entra in un byte 
75 if x < 256 then 85 
77 rem inserisce l'eccesso e setta 
78 rem il primo bit di v+16 
80 poke v,x-256: poke v+16,1: goto 90 
82 rem se invece x entra in un byte 
83 rem inserisce tutto e resetta 
84 rem il primo bit di v+16 
85 poke v,x: poke v+16,0 
90 poke v+1,y 
100 nx = x + dx 
110 if nx >= 0 and nx <= 344 then 130 
120 dx = -dx:nx = x + dx 
130 x = nx 
200 goto 75 
500 data 255,255,255 
501 data 128,0,1
502 data 128,0,1
503 data 128,0,1 
504 data 128,0,1 
505 data 128,0,1 
506 data 128,0,1 
507 data 128,0,1 
508 data 128,0,1 
509 data 128,0,1 
510 data 128,0,1 
511 data 128,0,1 
512 data 128,0,1
513 data 128,0,1
514 data 128,0,1 
515 data 128,0,1
516 data 128,0,1 
517 data 128,0,1 
518 data 128,0,1 
519 data 128,0,1
520 data 255,255,255
521 data 0

Altre amenità varie

L’origine di uno sprite è il pixel in alto a sinistra.
L’area di posizionamento degli sprite è più ampia dell’area di visualizzazione (320x200), prima e dopo gli 
sprite, letteralmente, spariscono sotto il bordo dello schermo.
I colori sono 16, codificati da 0 a 15.

